Scheme Notes 03

Geoffrey Matthews

Department of Computer Science
Western Washington University

January 26, 2018

Recursion *vs.* Tail-recursion

$$a^b = \begin{cases} 1 & \text{if } b = 0 \\ a(a^{b-1}) & \text{otherwise} \end{cases}$$

```
(define pow-rec
  (lambda (a b)
    (if (zero? b)
        1
        (* a (pow-rec a (- b 1)) ))))
```

Recursion *vs.* Tail-recursion

$$a^b = \begin{cases} 1 & \text{if } b = 0 \\ a(a^{b-1}) & \text{otherwise} \end{cases}$$

```
(define pow-rec
  (lambda (a b)
    (if (zero? b)
        1
        (* a (pow-rec a (- b 1)) ))))

(define pow-iter
  (lambda (a b)
    (define loop
      (lambda (b product)
        (if (zero? b)
            product
            (loop (- b 1) (* a product)) )))
    (loop b 1)))
```

Named `let`

```
(define pow-iter
  (lambda (a b)
    (define loop
      (lambda (b product)
        (if (zero? b)
            product
            (loop (- b 1) (* a product)))))
    (loop b 1)))


(define pow-iter-2
  (lambda (a b)
    (let loop ((b b) (product 1))
      (if (zero? b)
          product
          (loop (- b 1) (* a product))))))
```

Fast recursion

```
(define pow-fast
  (lambda (a b)
    (cond ((zero? b) 1)
          ((even? b) (sqr (pow-fast a (/ b 2))))
          (else (* a (pow-fast a (- b 1)))))))
```

Lists

A **list** is either:

1. the **empty list**, or
2. **an item** and a **list**

Lists in Scheme:

1. The empty list in Scheme: `'()`
2. Create a list from 3 and the empty list:
   `(cons 3 '()) ⇒ (3)`
3. Create the list `(4 7 2)`:
   `(cons 4 (cons 7 (cons 2 '()))) ⇒ (4 7 2)`
4. Shorthand for long lists: `(list 4 7 2) ⇒ (4 7 2)`

Lists in Scheme:

1. The empty list in Scheme: `'()`
2. Create a list from 3 and the empty list:
   `(cons 3 '()) ⇒ (3)`
3. Create the list `(4 7 2)`:
   `(cons 4 (cons 7 (cons 2 '()))) ⇒ (4 7 2)`
4. Shorthand for long lists: `(list 4 7 2) ⇒ (4 7 2)`
5. Using quote: `'(4 7 2) ⇒ (4 7 2)`
   `'(+ 4 7 2) ⇒ (+ 4 7 2)`
   `'(a b c) ⇒ (a b c)`
   `(a b c) ⇒ error`
   `(+ 4 7 2) ⇒ 13`
   `'(list (+ 2 2) 7 2) ⇒ (list (+ 2 2) 7 2)`
   `(list (+ 2 2) 7 2) ⇒ (4 7 2)`

Pictures of Lists



- The empty list has predicate `empty?`.
- The first pointer in a cons cell is called `car`.
- The second pointer in a cons cell is called `cdr`.
- Run `boxarrow.rkt` for pictures.

An improper list results in a dot:

- `(cons 4 8)` ⇒ `(4 . 8)`
- `(cons 3 (cons 5 (cons 7 9)))` ⇒ `(3 5 7 . 9)`
- Run `boxarrow.rkt` for pictures.



length

length

```
(define (length lst)
  (if (empty? lst)
      0
      (+ 1 (length (cdr lst)))))
```

nth

nth

```
(define (nth lst n)
  (cond ((empty? lst) '())
        ((= n 0) "Not defined")
        ((= n 1) (car lst))
        (else (nth (cdr lst) (- n 1)))))
```

last

last

```
(define (last lst)
  (cond ((empty? lst) '())
        ((empty? (cdr lst)) (car lst))
        (else (last (cdr lst)))))
```

scale-list

scale-list

```
(define (scale-list lst n)
  (if (empty? lst)
      '()
      (cons (* n (car lst))
            (scale-list (cdr lst) n))))
```

increment-list

increment-list

```
(define (increment-list lst)
  (if (empty? lst)
      '()
      (cons (+ 1 (car lst))
            (increment-list (cdr lst)))))
```

map

map

```
(define (map lst op)
  (if (empty? lst)
      '()
      (cons (op (car lst))
            (map (cdr lst) op))))
```

scale-list using map

scale-list using map

```
(define (scale-list lst n)
  (map lst (lambda (x) (* n x))))
```

increment-list using map

increment-list using map

```
(define (increment-list lst)
  (map lst (lambda (x) (+ x 1))))
```

append

append

```
(define (append lst1 lst2)
  (if (empty? lst1)
      lst2
      (cons (car lst1)
            (append (cdr lst1) lst2))))
```

remove

remove

```
(define (remove n lst)
  (cond ((empty? lst) '())
        ((= n (car lst)) (remove n (cdr lst)))
        (else (cons (car lst)
                    (remove n (cdr lst))))))
```

Trees

Trees

```
(define tree1 (cons (list 1 2) (list 3 4)))

(define tree2 (list (list 1 2)
                    (list 3
                          (list 4 5 6))
                    (list 7 8)))
```

- Run `boxarrow.rkt` for pictures.



count-leaves

count-leaves

```
(define (count-leaves tree)
  (cond ((empty? tree) 0)
        ((not (pair? tree)) 1)
        (else (+ (count-leaves (car tree))
                 (count-leaves (cdr tree))))))
```

fringe

fringe

```
(define (fringe tree)
  (cond ((empty? tree) '())
        ((not (pair? tree)) (list tree))
        (else (append (fringe (car tree))
                      (fringe (cdr tree))))))
```

sum-fringe

sum-fringe

```
(define (sum-fringe tree)
  (cond ((empty? tree) 0)
        ((number? tree) tree)
        (else (+ (sum-fringe (car tree))
                 (sum-fringe (cdr tree))))))
```

map-tree

map-tree

```
(define (map-tree tree op)
  (cond ((empty? tree) '())
        ((number? tree) (op tree))
        (else (cons (map-tree (car tree) op)
                    (map-tree (cdr tree) op)))))
```

scale-tree using map-tree

scale-tree using map-tree

```
(define (scale-tree tree factor)
  (map-tree tree (lambda (x) (* x factor))))
```

increment-tree using map-tree

increment-tree using map-tree

```
(define (increment-tree tree)
  (map-tree tree inc))
```